
Programacion ActiveX con Visual Basic
8 Temas de interés

Antes de adentramos en la seccion del libro dedicada exclusivamente al desarrollo de controles ActiveX, no estd de mas
repasar algunos temas de interés relacionados con ciertas novedades incluidas en la version 5.0 de Visual Basic. Quizd ya
conozca estos temas, pero en cualquier caso nunca estd de mas dar un pequefio repaso.

Algunos de los puntos que se van a estudiar ya se han usado en capitulos previos, como es el caso de las enumeraciones,
mientras que otros son nuevos. Veamos detalladamente como aprovecharlos y utilizarlos en la creacién de nuestros
propios componentes.

Enumeraciones

Ya hemos usado enumeraciones en algunos de los ejemplos desarrollados en capitulos previos, aunque sin detenernos a
estudiar su sintaxis y funcionamiento. La declaracién de una enumeracion es analoga a la de un tipo definido por un
usuario, si bien en este caso los miembros son sélo constantes representativas de los valores que podria tomar una
variable declarada de ese tipo. La siguiente es la definicién tipica de una enumeracion:

Public Enum DiaSemana
Lunes
Martes
Miercoles
Jueves
Viernes
Sabado
Domingo

End Enum

Dim Dia As DiaSemana

La variable Dia es del tipo DiaSemana, una enumeracién cuyos valores se representan mediante las constantes Lunes,
Martes, ... Domingo. En realidad, internamente Dia es una variable de tipo Long, por lo que seria posible asignarle
cualquier valor entero que estuviese en el rango de este tipo numérico. Una asignacién como la siguiente, por ejemplo,
no generaria error alguno:

Dia = 12

Al definir una enumeracién es posible asignar valores explicitamente a las constantes que la componen. Para ello bastara
con disponer detras del nombre de la constante el operador =y el valor a asignar. En la siguiente enumeracion, por
ejemplo, se asigna de forma explicita un valor diferente a cada una de la constantes.

Public Enum Color
Rojo = vbRed
Azul vbBlue
Verde = vbGreen
End Enum

Si tras asignar un valor a una constante se omite el valor de las siguientes, Visual Basic les asignard automaticamente el
valor de la constante anterior incrementado en una unidad.

Por supuesto las enumeraciones pueden ser privadas, en cuyo caso su uso se restringira sélo al interior del médulo en que
se haya declarado.

Ventajas de usar enumeraciones

Si, como se acaba de decir, es posible asignar cualquier valor a una variable cuyo tipo es una enumeracion, ; qué ventajas
tiene usarlas? ;jNo seria lo mismo declarar la variable de tipo Long y usar constantes definidas en la forma tradicional,
con la palabra clave Const? Si el uso de la variable y las constantes va a ser interno a nuestro componente, lo cierto es
que no habra mucha diferencia entre usar un método u otro, si exceptuamos que la declaracién de una variable con el tipo
DiaSemana define mejor los valores que puede almacenar. Pero si, por el contrario, la variable va a ser usada como

almacenamiento de una propiedad o tipo para los métodos de acceso correspondientes, las enumeraciones tienen visibles
ventajas.

Si definimos una propiedad con el tipo Long, durante el disefio serd posible asignarle cualquier valor usando el Inspector
de propiedades. Por el contrario, si definimos una enumeracién y la usamos como tipo, el Inspector de objetos mostrara
una lista desplegable con los valores posibles, no permitiendo al usuario del control que asigne valores invalidos. Cuando
en el Editor de codigo se haga referencia a la propiedad para asignarle un valor, Visual Basic, en caso de que sea ¢l el
contenedor del control, mostrara automaticamente una lista desplegable con la enumeracién de valores. Légicamente sera
posible asignar cualquier otro tecleandolo, ahi es donde el método de asignacién de la propiedad escrito por nosotros
entra en accién, rechazando cualquier valor no vélido.

A pesar de que solo lo dicho hasta ahora ya justificaria el uso de las enumeraciones en sustitucion de constantes
independientes, hay ventajas adicionales al usar esta nueva caracteristica del lenguaje afiadida en la versién 5.0. Las
constantes definidas en un médulo de clase que forma parte de un componente ActiveX, son sustituidas automaticamente
por los valores correspondientes. Es decir, las constantes en si no existiran en el componente, son sélo un recurso para
facilitar la codificaciéon. Una enumeracion, por el contrario, pasard a formar parte del componente ActiveX, de tal forma
que leyendo su libreria de tipos sera posible recuperar la informacién relativa a la enumeracién. En realidad es este
hecho, que la enumeracion se encuentre en la libreria de tipos ActiveX, el que permite que Visual Basic u otro
contenedor muestre los valores correctos en tiempo de disefio, en el Inspector de propiedades o un elemento equivalente
en otro entomo.

Por lo tanto, siempre que vaya a definir una propiedad que pueda tomar varios valores numéricos asociados a un rango
discreto de posibilidades, en lugar de usar el tipo Long o el tipo Integer defina una enumeracion y tisela como tipo. Los
beneficios obtenidos justifican este trabajo adicional.

Colecciones

La coleccién no es un elemento nuevo en Visual Basic, aunque su funcionalidad se ha ido ampliando con cada nueva
version. Inicialmente existian colecciones predefinidas que se podian usar, por ejemplo, para acceder a los formularios o
controles. Posteriormente, al aparecer el objeto Collection, se abrio la puerta al uso de las colecciones para almacenar
cualquier serie de datos que pudiera sernos util sin las limitaciones propias de las matrices. A partir de la versién 5.0 se
facilita la creacion de colecciones propias, con una funcionalidad idéntica a la del propio objeto Collection.

Seguramente estard pensando que ya en versiones anteriores de Visual Basic, como la 4.0, usted podia crear sus propias
clases de colecciones. Es cierto, pero existian limitaciones. Una de las ventajas del uso de las colecciones es que es
posible recorrerlas sin necesidad de conocer el nimero de elementos existentes ni usar indices, gracias a la construccion
For Each objeto In. Esta modalidad de bucle, sin embargo, no funciona con ningun otro objeto, sélo con el tipo
Collection. En la version 5.0, no obstante, si es posible conseguir la misma funcionalidad en cualquier otra clase de
coleccién creada por nosotros.

Objetos Collection publicos

Si est4 disefiando un componente que cuenta con una coleccién de objetos cualesquiera y, por flexibilidad, quiere que los
usuarios puedan acceder a esa coleccion, accediendo a sus elementos, afiadiendo o eliminando, una de las posibilidades
que tiene es hacer publico el objeto Collection. Para ello bastaria con declarar una variable publica del tipo Collection,
creando la coleccidn directamente en la propia declaracién o bien en un evento de inicializacion.

Este método, no cabe duda, es el mas sencillo posible. Ademas es muy flexible, puesto que el usuario de nuestro
componente podra acceder a la coleccion sin limitacién alguna, pudiendo acceder a los elementos y manipularlos. Esta
flexibilidad, sin embargo, es un arma de doble filo, puesto que el usuario podria insertar en la coleccién objetos de tipos
no validos para nosotros o valores inapropiados. No podemos controlar esto, puesto que los métodos que facilitan las
operaciones sobre la coleccidn son propios del objeto Collection, no nuestros.

Clases de colecciones propias

Para evitar los problemas que se derivan de hacer publico un miembro de datos de un objeto, lo mejor es crear una clase
implementando las propiedades y métodos que se precisen. De esta forma, mediante los métodos de acceso
correspondientes, podremos evitar que el usuario introduzca valores erréneos en la coleccion.

Con el fin de mantener la coherencia con cualquier otra coleccidn, nuestra clase deberd contar con los métodos Add'y
Remove, asi como con las propiedades Count e Item, siendo ésta tltima la propiedad predeterminada. De esta forma
podriamos crear, por ejemplo, una coleccioén que tan solo aceptase dias de la semana, tal y como se muestra a

continuacion.

Public Enum DiaSemana ' Enumeracidén de valores posibles
Lunes
Martes
Miercoles
Jueves
Viernes
Sabado
Domingo
End Enum

' Objeto privado para mantener la coleccidn
Private PColeccion As New Collection

' Propiedad sélo de lectura que devuelve el numero de elementos
Public Property Get Count () As Long

Count = PColeccion.Count
End Property

' Propiedad sélo de lectura que devuelve un elemento
Public Property Get Item(Indice As Long) As DiaSemana
If Indice > 0 And Indice <= PColeccion.Count Then _
Item = PColeccion(Indice)
End Property

' Método para afladir un nuevo elemento
Public Sub Add(Elemento As DiaSemana)
If Elemento >= Lunes And Elemento <= Domingo Then _
PColeccion.Add Elemento
End Sub

' Método para eliminar un elemento existentes
Public Sub Remove (Indice As Long)
If Indice > 0 And Indice <= PColeccion.Count Then _
PColeccion.Remove Indice
End Sub

Suponiendo que este codigo se aloja en un modulo de clase llamado PAXDiasSemana, podriamos crear un objeto de
dicha clase y usar sus propiedades y métodos para afiadir, examinar y eliminar elementos. Légicamente también
podriamos usar la clase como tipo de propiedad de un control, teniendo la seguridad de que el usuario no va a poder
introducir en la coleccién valores no vélidos. El control PAX Control es un ejemplo simple que implementa una
propiedad, llamada Dias, que facilita el acceso a la coleccion privada PDias, como puede verse a continuacion.

' Variable privada para almacenar la colecciédn
Private PDias As New PAXDiasSemana

' Método de lectura de la propiedad Dias
Public Property Get Dias () As PAXDiasSemana
Set Dias = PDias
End Property

Este control, l6gicamente, no es de suma utilidad. Insertando uno en un formulario podriamos afiadir, eliminar y acceder
a los elementos de la propiedad Dias como si de cualquier otra coleccion se tratase. Sin embargo, al intentar usar la
construccidén For Each obtendriamos un error.

' Al pulsar el botdn
Private Sub Commandl_Click ()
With PAXControll
.Dias.Add Lunes ' Afadimos dos elementos
.Dias.Add Viernes
.Dias.Remove 1 ' eliminamos uno
Cls

Print "Contador = " & .Dias.Count ' mostramos el numero de elementos
Dim Dia As Variant
For Each Dia In .Dias
Print Dia ' y el contenido de cada uno
Next
End With
End Sub

La interfaz IEnumVARIANT

,Qué tiene de especial el tipo Collection? ;Por qué se puede usar con la construccidén For Each? La respuesta estd en
que, a diferencia de cualquier clase que nosotros podamos definir, Collection implementa, entre otras, la interfaz
IEnumVARIANT. Esta es una interfaz estindar COM que facilita la enumeracion de valores por parte de un objeto, sin
importar su tipo puesto que se usa siempre el tipo Variant.

La instruccién For Each lo que hace es obtener la interfaz IEnumVARIANT del objeto que se facilita a continuacion,
llamando para ello al método oculto NewEnum. So6lo los objetos Collection cuentan con este método, pero esta claro que
nosotros podemos afiadirle uno a nuestra propia clase. El problema estriba en que nuestra propia clase no implementa la
interfaz IEnumVARIANT, por lo que de implementar el método NewEnum jqué devolveriamos?

Todo objeto Collection cuenta con el método NewEnum, que al estar oculto recibe el nombre NewFEnum. Podemos
delegar nuestro método NewEnum en el de la coleccion interna que estamos usando, al igual que hemos delegado muchas
funciones en el objeto UserControl en capitulos previos. Para ello bastard con que nuestro método New Enum devuelva el
valor devuelto a su vez por el método NewEnum de dicha coleccién.

Como se acaba de decir, el método NewEnum del objeto Collection esta oculto, por lo que su nombre es  NewFEnum. Ese
subrayado inicial impide que podamos, en Visual Basic, referirnos directamente al identificador. No podemos escribir
una referencia como la siguiente:

Set NewEnum = PColeccion._NewEnum

Para poder usar estos caracteres invalidos en un identificador, es necesario delimitar éste mediante unos corchetes. De
esta forma, nuestro método New Enum podria quedar de la siguiente forma.

' Devuelve la interfaz que permite recorrer la enumeracién
Public Property Get NewEnum() As IEnumVARIANT
Set NewEnum = PColeccion. [_NewEnum]
End Property

Escrito este método deberemos usar la ventana de atributos de procedimiento para establecer algunos parametros.
Nuestro método ha de ser oculto, por lo que hay que activar la opcidén Ocultar este miembro, tal y como se puede
apreciar en la figura 8.1. Ademas tendremos que asignar el identificador -4 al método, ya que For Each no usa un
método llamado New Enum, sino un método con ese identificador.

Figura 8.1: Atributos del método NewEnum

Dados todos estos pasos ya podremos usar una coleccién PAXDiasSemana como cualquier otra coleccién, siendo
posible enumerar sus elementos mediante la construccion For Each. El proceso es algo "retorcido”, por decirlo de alguna
forma, pero no tenemos otra opcidn para conseguir el resultado que queremos, a no ser que deseemos implementar
nosotros directamente la interfaz JEnumVARIANT.

Direcciones de funciones

A pesar de que la posibilidad de acceder a las funciones de 1a API de Windows desde Visual Basic siempre ha estado
presente, lo cierto es que hasta la versién 5.0 no se ha incorporado al lenguaje un elemento "fundamental" para muchos
programadores, como es la posibilidad de obtener la direccién de una funcién. ;Para qué necesitamos esta direccién?
Existen multitud de casos en los que puede ser imprescindible. Cualquier funcién Windows que requiera el uso de una
funcioén callback, por ejemplo, precisara la direccion de dicha funcién. Asimismo serd imprescindible obtener la
direccién de una funcién si queremos, por ejemplo, subclasificar una ventana.

Una de las novedades de la version 5.0 de Visual Basic es el operador AddressOf, que dispuesto delante del nombre de
una funciéon devuelve su direccidn. Esta operacién, aparentemente tan simple, abre la puerta para que los programadores
Visual Basic puedan efectuar muchas operaciones que hasta ahora estaban fuera de su alcance.

Limitaciones de AddressOf

El uso de punteros a funciones es potencialmente muy peligroso, los programadores de lenguajes como C/C++y Pascal
lo saben bien. Es muy facil asignar una direccién errénea vy bloquear la aplicacién o, en ocasiones, incluso el sistema
completo. Seguramente esta es la razén de que el operador AddressOf esté tan limitado, a pesar de lo cual, como se ha
dicho antes, puede ser muy util.

Este operador tan sélo puede ser usado como pardmetro en la llamada a un método o funcién, es decir, no se puede
almacenar el valor devuelto por AddressOf directamente en una variable, puesto que una sentencia como la siguiente es
invalida.

Dim Direccion As Long
Direccion = AddressOf MiFuncion

Tras escribir la sentencia de asignacién y pulsar <Intro> vera que automaticamente Visual Basic avisa que la expresién

no es valida. Si queremos almacenar una direccién en una variable, tendremos que escribir una funcién como la
siguiente:

Function DireccionFuncion (Dir As Long) As Long
DireccionFuncion = Dir
End Function

Direccion = DireccionFuncion (AddressOf Enumera.EnumeraVentanas)

En este caso la variable Direccion si almacena la direccion de la funcién, que ha sido pasada como pardmetro a la
funciéon DireccionFuncion. Esta funcion la devuelve sin maés, no es necesario realizar operacion alguna.

La funcion que se utiliza como pardmetro del operador AddressOfha de alojarse siempre en un mddulo de codigo
estandar. No es posible obtener la direccion de un método de un objeto, por ejemplo un control, puesto que un elemento
como ese no es una funcién "normal”, sino una entrada en una tabla de direcciones de una interfaz COM, tal y como
vimos en el capitulo correspondlente.

Este hecho, que la funcién cuya direccién se va a obtener deba encontrarse en un modulo estandar, complica su uso en
proyectos como controles y componentes ActiveX, en los cuales la encapsulacién es un elemento de vital importancia.

Un control enumerador de ventanas

Veamos con un pequefio ejemplo como podemos combinar un moédulo estandar con un control de usuario. La finalidad
sera construir un control, no visible durante la ejecucion, que contard con un método y un evento. E1 método, al que
llamaremos EnumeraVentanas, se encargard de obtener el nombre de todas las ventanas de primer nivel que haya
abiertas, generando un evento Ventana por cada una de ellas. E1 método asociado a ese evento recibird como parametro
el titulo de la ventana.

Para obtener una lista de las ventanas podemos usar distintas funciones de la API de Windows. En este caso vamos a
utilizar EnumWindows, funcion a la que hay que pasar dos parametros: la direccion de una funcién callback y un dato de
usuario que puede ser cualquier informacion que nos interese pasar a esa funcion callback. Tras llamar a EnumWindows,
se pondrd en marcha un proceso por el que Windows llamara a la funcién cuya direccién se ha pasado tantas veces como
ventanas haya. En cada llamada se facilita como primer parametro el identificador de una ventana y, como segundo, el
dato que nosotros mismos hubiésemos pasado en la llamada a EnumWindows.

Teniendo un identificador de ventana, para recuperar su nombre usaremos la funcion GetWindowText. Al llamarla
entregaremos como primer parametro el identificador, como segundo una cadena y como tercero un entero indicando el
espacio disponible en dicha cadena. Esta funcién devolvera como resultado un nimero comunicando la longitud de la
cadena devuelta, o cero en caso de que la ventana no tenga nombre o no exista.

Puesto que la funcién callback tendra que encontrarse en un médulo estandar, deberemos establecer de alguna forma una
comunicacion entre ella y el control que vamos a disefiar. Para ello declararemos en dicho médulo una variable que nos
permitird mantener una referencia al control que lo estd usando. Cuando se cree un PAXEnumeraVentanas ¢] mismo
llamar4 al método Asocia que hay en el médulo, facilitando como pardmetro una referencia a si mismo. En el momento
en que el control se vaya a destruir, se llamara al método Cancela para eliminar la asociacion. El codigo del control es el
que se puede ver a continuacion.

Option Explicit

' Evento que generard el control
Event Ventana (Titulo As String)

' Este método publico serd llamado desde el cdédigo del mdédulo estandar
Public Sub TituloVentana(Titulo As String)

RaiseEvent Ventana (Titulo)
End Sub

' Este es el Unico método accesible para el usuario
Public Sub EnumeraVentanas ()

Enumera.ListaVentanas
End Sub

' Al crear el control

Private Sub UserControl_TInitialize ()
Enumera.Asocia Me ' lo asociamos con el mdédulo

End Sub

' Al destruir el control
Private Sub UserControl_Terminate ()
Enumera.Cancela ' cancelamos la asociaciédn
End Sub

El método publico TituloVentana estd pensado para ser llamado desde el mddulo de codigo estdandar cada vez que se
obtenga un nombre de ventana. Sin embargo, este método no deberia de estar accesible para el usuario final, por lo que
usando la ventana de propiedades de procedimiento lo haremos oculto. Que el método no aparezca en el Examinador de
objetos ni en la lista de miembros cuando se esta escribiendo co6digo, no significa que no se le pueda llamar. En el cédigo
del modulo estandar, mostrado seguidamente, se llama a TituloVentana por cada titulo de ventana que se recupera. El
control traduce esta llamada en un evento Ventana.

' Control asociado a la operacidn
Private Objeto As PAXEnumeraVentanas

' Esta es la funcidén callback, que recibe el identificador de la
' ventana y un pardmetro adicional
Private Function EnumeraVentanas (ByVal hwnd As Long, ByVal LPARAM As Long) As Lon
Dim Titulo As String * 256
' si podemos obtener el titulo
If GetWindowText (hwnd, Titulo, 255) <> 0 Then _
Objeto.TituloVentana Titulo ' lo pasamos al control
EnumeraVentanas = True ' y seguimos con la enumeraciédn
End Function

' Al llamar a este procedimiento
Public Sub ListaVentanas ()
If TypeName (Objeto) = "PAXEnumeraVentanas" Then _
EnumWindows AddressOf EnumeraVentanas, 0 ' e iniciamos la enumeracidn
End Sub

' Este método asocia el control con el mdédulo
Public Sub Asocia (L As PAXEnumeraVentanas)
Set Objeto = L
End Sub

' Este método cancela la asociacidn
Public Sub Cancela /()

Set Objeto = Nothing
End Sub

Tras dar el valor True a la propiedad InvisibleAtRuntime del control, ya estamos preparados para comprobar su
funcionamiento. Bastara con insertar en un formulario una lista, un botén y un PAXEnumeraVentanas. Al pulsar el
botén se limpiard la lista y se llamard al método EnumeraVentanas de nuestro control. Cada vez que se reciba un
evento Ventana se afiadira el titulo a la lista. La figura 8.2 muestra el programa en funcionamiento. Antes de ejecutar
cualquier programa en el que se utiliza el operador AddressOf siempre es recomendable guardar el trabajo, ya que un
fallo podria hacer que todo se perdiese.

Figura 8.2: Lista de ventanas recuperada por el control PAXEnumeraVentanas

Observe que la funcion EnumeraVentanas devuelve siempre el valor True, ya que de lo contrario el proceso se
detendria de forma inmediata, obteniéndose tan sélo el identificador de una ventana. Fijese también en la comprobacién
que se realiza, en el procedimiento ListaVentanas, justo antes de llamar a EnumWindows. En caso de que la variable
Objeto no tenga una referencia a un PAXEnumeraVentanas no se realiza el proceso de enumeracion.

Los problemas de las variables publicas

Como ya sabe, las variables declaradas en el interior de un médulo de clase o un control no existen por si solas, sino que
forman parte de un objeto. De esta forma, hasta que no se crea el objeto la variable no existe. No es posible acceder
directamente a la variable, sino que hay siempre que componer una referencia formada por el objeto y el nombre de la
variable.

Una variable en un modulo de codigo estandar, a pesar de que sea una variable privada, es un espacio de memoria global.
Esto quiere decir que es compartida por todas las copias de los objetos que puedan crearse, en este caso por todos los
controles PAXEnumeraVentanas. Puede hacer una prueba muy simple: habra el formulario del programa anterior e
inserte una nueva lista y un nuevo PAXEnumeraVentanas, de tal forma que al pulsar el botén se inicien dos
enumeraciones, dirigiendo la salida de una a una lista y la de otra a la segunda lista, tal y como se deduce del codigo
siguiente:

Private Sub Commandl_Click ()
Listl.Clear
List2.Clear
PAXEnumeraVentanasl.EnumeraVentanas
PAXEnumeraVentanas?2.EnumeraVentanas
End Sub

Private Sub Form_Resize ()
Listl.Width = ScaleWidth - Listl.Left * 2
Commandl .Width = Listl.Width

End Sub

Private Sub PAXEnumeraVentanasl_Ventana (Titulo As String)
Listl.AddItem Titulo
End Sub

Private Sub PAXEnumeraVentanas2_Ventana (Titulo As String)
List2.AddItem Titulo
End Sub

Al ejecutar el programa obtendr4 un resultado similar al de la figura 8.3. En una de las listas aparecen todas las entradas
duplicadas, mientras que la otra permanece vacia. ;jPor qué ocurre esto? Analicemos detenidamente el proceso que se
sigue hasta llegar aqui. Al crear la primera copia del control PAXEnumeraVentanas éste se asocia con el médulo
estandar, que guarda la referencia en la variable global Objeto. Acto seguido se crea la segunda copia del control, que
llama al método Asecia para enlazarse con el médulo global. La asignacién de la nueva referencia a la variable Objeto

hace que la anterior referencia se pierda. Cuando el primer control llama al procedimiento EnumeraVentanas, el destino
de la notificacién es ya la segunda copia del control, puesta que ésta es la referencia almacenada en la variable Objeto.

im. Lista de ventanas [_ |O]

Explorando - 0:SLibroz \PactivexconyB EKEiempIDs'\DEHPﬁ}:j
Ewplaranda - D:\LibrozhPactivesiconty'B BAE jemplozh 084 P
File

File

Hidden Congale of Wiiw WDk J
Hidden Console of W0 WD

Lizta de wentanas

Lizta de ventanas

MCI command handling window

FCl command handling window

Microgoft \word - 03.000C

Micrazaft Wward - 08.00C

k5 _wWebcheck kM anitor _*j

Enumera

Figura 8.3: Resultado al usar dos copias del control PAXEnumeraVentanas

Cuando el segundo control se destruya, produciéndose el evento Terminate, tendra lugar una llamada al procedimiento
Cancela del modulo estandar. En ese momento se asigna el valor Nothing a la variable Objeto. La primera copia del
PAXEnumeraVentanas, sin embargo, cree que aun permanece asociado, por lo que podria llamar de nuevo a
ListaVentanas. Por suerte en ese procedimiento hemos incluido una comprobacién que evita la llamada a
EnumWindows en caso de que la referencia Objeto sea Nothing, de lo contrario se produciria un error.

Encapsulaciéon de un médulo estandar

Para conseguir un correcto funcionamiento en cualquier caso, debemos encontrar una forma de encapsular el c6digo del
moédulo estandar. Lo mas facil seria alojar ese codigo en un médulo de clase, pero ya hemos dicho antes que el operador
AddressOf exige que la funcién cuya direccion se quiere obtener esté en un modulo estandar.

(Como funciona un modulo de clase? Cada vez que se crea un objeto de una clase se asigna el espacio necesario para
alojar las variables definidas a nivel de modulo, de tal forma que cada objeto, a pesar de compartir el mismo c6digo
ejecutable con los demas, dispone de su conjunto particular de variables, independientes de las de los demas objetos. Por
lo tanto, de lo que se trata es de conseguir que cada objeto asociado a un modulo estandar cuente con su propia copia de
las variables. ;,Cémo conseguir esto? Lo cierto es que hay muchas formas, casi tantas como podamos imaginar. En el
caso que nos ocupa, sin embargo, lo mas facil es eliminar la variable global.

La funcion EnumWindows acepta como segundo pardmetro un valor cualquiera, valor que sera posteriormente facilitado
a la funcion callback. Inicialmente dicho pardmetro esta definido como Long y para ser pasado por valor, pero nada nos
impide modificar esta definicién para poder pasar una referencia al control PAXEnumeraVentanas asociado. Esta
posibilidad, de la cual disponen muchas funciones de la API de Windows, nos facilitard mucho el trabajo.

Comenzaremos por modificar el cddigo del control, eliminando los métodos correspondientes a los eventos Initialize y
Terminate. Serd al llamar al procedimiento ListaVentanas cuando facilitemos una referencia al propio control, tal y
como puede verse en el cddigo siguiente.

' Evento que generard el control
Event Ventana (Titulo As String)

' Este método publico serd llamado desde el cdédigo del mdédulo estandar
Public Sub TituloVentana (Titulo As String)

RaiseEvent Ventana (Titulo)
End Sub

' Este es el Unico método accesible para el usuario
Public Sub EnumeraVentanas ()

Enumera.ListaVentanas Me
End Sub

Este es todo el codigo del control, que se ha simplificado apreciablemente. También el codigo del modulo estandar se va
areducir. Ya no son precisos los procedimientos Asocia y Cancela, que pueden ser eliminados. Tampoco necesitamos
ya la variable Objeto, puesto que el parametro que se pasa a ListaVentanas no va a almacenarse, sino que se entregara
como pardmetro a la funcion EnumWindows. Esta, a su vez, nos lo devolveré en cada llamada a la funcion
EnumeraVentanas, cuyo cddigo hemos modificado para que LPARAM no sea de tipo Long, sino una referencia a un
control PAXEnumeraVentanas2. Usando esta referencia no tenemos mas que hacer la llamada para notificar la
recepcién de un titulo de ventana.

' Para obtener una enumeracién de todas las ventanas de primer nivel
Private Declare Function EnumWindows Lib "user32"

(ByVal lpEnumFunc As Long,
ByRef LPARAM As Any) As Long

' Esta es la funcidn callback, que recibe el identificador de la ventana y un
Private Function EnumeraVentanas (ByVal hwnd As Long,
ByRef LPARAM As PAXEnumeraVentanas2) As Long
Dim Titulo As String * 256
' si podemos obtener el titulo
If GetWindowText (hwnd, Titulo, 255) <> 0 Then _
LPARAM.TituloVentana Titulo ' lo pasamos al control
EnumeraVentanas = True ' y seguimos con la enumeracién
End Function

' Al llamar a este procedimiento
Public Sub ListaVentanas (L As PAXEnumeraVentanas2)

EnumWindows AddressOf EnumeraVentanas, L ' e iniciamos la enumeracidn
End Sub

Hechas estas modificaciones podemos realizar la misma prueba anterior, usando dos controles PAXEnumeraVentanas?2
para llenar dos listas con los nombres de las ventanas. En este caso no hay ningun problema, cada lista recibe los
elementos que le corresponden.

. Lista de ventanas [_ (O]

Contador de la bateria -
CPAXE numiventanas? - Microgzaft Yisual Basic [ejecucion]

CP&xE numi'entanas? - Microzoft Wizual Bazic [ejecucion] - [Enumera [Cadiga]]

DOE Server Windaw

DOE Server Window

DDE Server Window

DOE Server Windaw

DOE Server Window

Esplaranda - D:\LibrozhPactivercontB 55E jemplog 084\Fae<E numeray'entanas?
Hidden Canzale of Wilw WDk

Lizta de wentanas

Ml command handling window

Micrasoft Word - 08.D0C hd

Enumera

Contador de la bateria -
CP&HE numtentanas? - Microzoft Yizual Basic [ejecucian]

CP&xE numi'entanas? - Microgzoft Visual Bazic [ejecucion] - [Enumera [Cadiga]]

DDE Server Window

DDE Server Windaw

DDE Server Window

DDE Server Window

DDE Server Windaw

Ewplaranda - D:\LibrozhPactivexconyB 55E jemplos 084PieE numera'entanas?
Hidden Conzole of W YD M

Lizta de wentanas

MCI| command handling window

Microsoft Word - 08.00C hd

Figura 8.4: Resultado usando el control PAXEnumeraVentanas?2

En posteriores ejemplos tendremos ocasién de conocer otras formas de encapsular el cédigo de un médulo estandar, de
tal modo que pueda ser utilizado de forma segura por cualquier numero de clientes de forma simultanea.

Subclasificacion

La posibilidad de obtener la direccioén de una funcién, mediante el operador AddressOf, no es util tan sélo en casos como
el visto en el punto anterior, sino que nos permite realizar tareas mucho m4s interesantes v que, hasta ahora, estaban
fuera del alcance de los programadores que utilizan Visual Basic. Una de esas tareas es la posibilidad de subclasificar
una ventana, interceptando todos los mensajes que recibe. Subclasificando una ventana se pueden realizar procesos que
no serian posibles de otra forma, salvo utilizando controles o librerias externas, pero también es muy facil provocar el
mal funcionamiento de un programa. Se trata, por lo tanto, de una posibilidad muy potente y peligrosa a la vez.

No solo los formularios son ventanas

El primer concepto que debemos tener claro es qué es una ventana en el sistema operativo Windows. Si le preguntamos a
cualquier usuario, posiblemente nos dird que una ventana es ese espacio en pantalla delimitado por un borde y en cuyo
interior se solicitan o muestran datos. Para la mayoria de los programadores de entornos de desarrollo rapido, como
Visual Basic, una ventana es un contenedor en el que se pueden alojar componentes, por lo que una ventana seria el
formulario. Para Windows, sin embargo, una ventana es una estructura de datos en memoria que define un posible
aspecto, ademas de contener un apuntador a una funcién que serd la encargada de gestionar los mensajes que se reciban.

Son ventanas, por lo tanto, los formularios, los botones, las listas y practicamente cualquier otro control visual salvo
excepciones. También son ventanas los controles creados por nosotros mismos. Lo unico que diferencia a estos
elementos es su clase, previamente registrada en Windows mediante RegisterClass o RegisterClassEx. Las clases de
ventana no tienen nada que ver con las clases de objetos Visual Basic. Una clase de ventana es simplemente una
estructura de datos que se registra con un cierto nombre, de tal forma que después es posible crear ventanas de esa clase
usando dicho nombre.

Procedimientos de ventana

Entre los miembros de la estructura de datos que define una clase de ventana, existe uno que contiene la direccién de una

funcién conocida habitualmente como "procedimiento de ventana". Dicha funcién es compartida, por regla general, por
todas las ventanas de esa clase. Cuando se crea una ventana, en ultima instancia con las funciones CreateWindow o
CreateWindowEx, Windows usa la informacion de la clase, previamente registrada, para saber qué procedimiento de
ventana tiene que utilizar.

La funcién que actiia como procedimiento de ventana recibird cuatro pardmetros: el identificador de la ventana a la que
se dirige el mensaje, el mensaje propiamente dicho y dos datos asociados a éste. Como funcién que es, el procedimiento
de ventana debera devolver un cédigo, cuyo significado depender4 directamente del mensaje que se haya recibido. Por lo
tanto, en Visual Basic podriamos implementar una funcién de proceso de mensajes tal y como se muestra en el siguiente
fragmento.

Function ProcesoMensajes (ByVal hWnd As Long, ByVal Mensaje As Long,
ByVal wParam As Long, ByVal lParam As Long) As Long

End Function

Los cuatro parametros se reciben por valor y son de tipo entero. El identificador de ventana es analogo a la propiedad
hWnd que tienen la mayoria de los controles Visual Basic, asi como los formularios. El mensaje se representa mediante
una constante, que podemos recuperar usando el Visor API. Los dos pardmetros restantes serdn datos asociados al
mensaje, su significado dependera de qué mensaje se reciba.

Subclasificar una ventana

Cuando creamos un programa usando Visual Basic, no tenemos que preocuparnos de registrar clases de ventanas, ni
siquiera de crear las propias ventanas, estas tareas las efectia automaticamente Visual Basic. No podemos, por lo tanto,
establecer una funcién nuestra como procedimiento de ventana en la propia clase de ventana.

Al crearse una ventana Windows asigna espacio en memoria para una estructura de datos privada, estructura que, entre
otros elementos, contiene la direccién del procedimiento de esa ventana en particular. Inicialmente la informacién
almacenada en la estructura de datos privada se copia de la estructura de la clase de ventana, pero Windows no impide
que se modifique. Disponiendo del identificador de la ventana, que podemos recuperar en cualquier momento de la
propiedad AWnd, lo inico que necesitamos es un medio para acceder a su estructura de datos privada y modificar la
direccién del procedimiento actual, estableciendo la de nuestra funcion.

Mediante la funcion SetWindowLong es posible modificar ciertos apartados de la estructura de datos privada de una
ventana, entre ellos la direccion del procedimiento de ventana. Al llamar a esta funcion deberemos pasar tres parametros:
el identificador de la ventana cuyos datos queremos alterar, una constante indicando qué dato de la estructura se va a
modificar y el nuevo valor a establecer. SetWWindowLong devuelve el antiguo valor que tuviese el dato modificado, lo que
nos permite conservarlo para una posterior restitucidén. En la tabla 8.1 se enumeran las constantes que identifican los
diferentes apartados que podemos modificar.

Tabla 8.1: Constantes para la funcion SetWindowLong

Constante Dato a modificar

GWL EXSTYLE Estilo extendido de la ventana

GWL STYLE Estilo basico de la ventana

GWL WNDPROC Direccion del procedimiento de la ventana
GWL HINSTANCE Identificador de la aplicacién

GWL ID Identificador de la ventana

GWL USERDATA Dato de usuario

Por lo tanto, suponiendo que tuviésemos un control y la funcién ProcesoMensajes anterior en un modulo de codigo
estandar, podriamos subclasificar el control con una sentencia como la siguiente.

Private ProcAnterior As Long

ProcAnterior = SetWindowLong (hWnd, GWL_WNDPROC, AddressOf ProcesoMensaijes)

Subclasificada la ventana, en este caso el control, todos los mensajes enviados a €l pasaran por nuestra funcién. Si no se
procesa ninguno de estos mensajes el control quedara inservible, ya que no sabra como dibujarse ni reaccionar a acciéon
alguna. Todos aquellos mensajes que no vayan a ser procesados por nuestro c6digo se deben pasar al anterior

procedimiento de ventana, para lo cual sera preciso usar la funcion CalllWindowProc. Al llamarla facilitaremos como
primer parametro la direccién del procedimiento, que hemos guardado en la variable ProcAnterior, y a continuacién los
mismos cuatro parametros que recibimos en ProcesoMensajes, en el mismo orden.

El encadenamiento de un procedimiento de ventana con el anterior es fundamental, ya que permite que una misma
ventana se subclasifique varias veces sin por ello modificar su funcionamiento. También es de vital importancia restituir
la anterior direccién en el momento en que nuestro programa o control se vaya a descargar. Con este fin usaremos de
nuevo la funcion SetWindowLong, pasando como tercer pardmetro la variable en la cual habiamos almacenado la
direccién original.

Un componente para uso de palancas de juegos

Veamos un ejemplo practico de como subclasificar una ventana, en este caso un control, creando un componente que nos
facilite el trabajo con palancas de juegos o joytiscks. Para utilizar este dispositivo existen una serie de funciones en la
API de Windows, mediante las cuales podemos saber cudntas palancas de juegos hay conectadas, obtener informacién
acerca de ellas, etc. Una palanca de juegos de tipo estandar dispone de entre dos y cuatro botones, cada uno de los cuales
puede estar pulsado o no. Asimismo, lo habitual es que tengan dos o tres ejes, a los que se suele denominar X, Yy Z.

En lugar de estar llamando peridédicamente a una funcién para saber si se ha pulsado o liberado un botén, o si se ha
registrado un movimiento en alguno de los ejes, podemos indicar a Windows que nos envie un mensaje cada vez que se
produzca alguno de esos eventos. Para poder recibir los mensajes necesitaremos subclasificar una ventana, tarea que
habria que realizar en cada programa en el que se fuese a utilizar una palanca de juegos.

Desarrollando un control se facilita el trabajo, puesto que toda la complejidad de la subclasificacion y gestion de los
mensajes se realiza en su interior. El usuario del control no tiene mas que insertarlo en un formulario, establecer
propiedades y responder a los eventos, sin mas.

Estructura del control

Este control, al que llamaremos PAXJoystick, serd no visible durante la ejecucién, por lo que daremos el valor 7rue a la
propiedad InvisibleAtRuntime. Durante el disefio aparecerd simplemente como un grafico, que asignaremos a las
propiedades Picture y ToolboxBimap. En la tabla 8.2 se resumen las propiedades y eventos que tendra.

Tabla 8.2: Propiedades y eventos del

control PAXJoystick
Miembro Comentario
. . Propiedad con el identificador del dispositivo a usar, podra ser PAXJoy1 o

UsarDispositivo
PAXJoy2

GenerarEventos Propiedad de tipo Boolean que indicar4 si hay que generar eventos o no

MovimientoMinimo Propleda_d_que especificara el movimiento minimo que se ha de registrar en
el dispositivo antes de producirse un evento

EstaConectado Propiedad sélo de lectu_ra que contendra 7rue si hay una palanca conectada
o False en caso contrario

PosX, PosY, PosZ Propiedades sélo de lectura con las coordenadas en los tres ejes

Botones Propiedad sélo de lectura con el estado de los botones

OnButtonUp Evento que se generara al liberarse un botén

OnButtonDown Evento que se generara al pulsarse un botén

OnMove Evento que se generara al registrar un movimiento de la palanca

Mediante la propiedad UsarDispositivo podremos elegir cual de los dos conectores de joystick queremos usar. Tras
hacer esta seleccion podemos consultar el valor de la propiedad EstaConectado, comprobando si hay o no una palanca
de juegos conectada. En caso afirmativo podemos consultar las propiedades PosX, PosY y PosZ para conocer la
posicién actual de la palanca en los tres ejes. Estos parametros suelen estar comprendidos entre 0 y 65535, aunque su tipo
es Long y, por lo tanto, el rango de valores podria ser mayor. La propiedad Botones devuelve un entero con el estado
actual de los botones. Para comprobar el estado de uno de ellos deberemos usar el operador logico And junto con una de
las constantes de la enumeracién PAXJoyBotones. Si damos el valor 7rue a la propiedad GenerarEventos, serd el
propio control el que genere eventos OnButtonUp, OnButtonDown y OnMove segiin la accidon que se detecte. La
mayoria de las palancas de juegos, sobre todo las analdgicas, no son capaces de mantener una posicion estable, por lo
que las coordenadas de los ejes pueden sufrir modificaciones aleatorias y continuas. Para evitar que esto se traduzca en
una generacién continua de eventos, es posible establecer un movimiento minimo mediante la propiedad

MovimientoMinimo.

Todo este funcionamiento se basa en el uso de una serie de funciones de la API de Windows, como son joyGetNumDevs,
joyGetPos, joySetCapture, joyReleaseCapture y joySetThreshold.

Para que el control pueda recibir los eventos generados por el dispositivo serd preciso hacer una subclasificacion, fin con
el cual tendremos que afiadir un médulo estandar al proyecto. En dicho médulo alojaremos la funcién que actuard como
procedimiento de ventana. El enlace entre el control y el médulo lo efectuaremos mediante una variable publica, que
almacenard una referencia al control subclasificado.

Inicializacion y destruccion del control

Comencemos por analizar el cédigo encargado de inicializar y destruir el control. Cuando se produzca el evento Initialize
el control acabara de crearse, siendo el momento apropiado para efectuar la subclasificacién. En el método asociado a ese
evento enlazaremos el control con €l modulo estandar, asignando la referencia Me a la variable ObjetoAsociado del
Modulo. Seguidamente usamos la funcién SetlWindowLong para subclasificar el control, estableciendo como nuevo
procedimiento de ventana la funcion ProcesoMensaje que hay en Modulo. La direccion del antiguo procedimiento la
conservaremos en la variable publica ProcedimientoAnterior, también definida en el moédulo estandar.

' Al inicializar el control
Private Sub UserControl_Initialize()
Set Modulo.ObjetoAsociado = Me
Modulo.ProcedimientoAnterior = _
SetWindowLong (hWnd, GWL_WNDPROC, AddressOf Modulo.ProcesoMensaije)
End Sub

Tras el evento Initialize serd preciso inicializar las propiedades, InitProperties, o leerlas, ReadProperties, segun que el
control se acabe de crear o se esté recuperando de un flujo de datos. En el primer caso asignamos a las propiedades
UsarDispositivo, GenerarEventos y MovimientoMinimo sus valores por defecto, mientras que en el segundo se
recuperan dichas propiedades usando los mismos valores como predeterminados. Observe que en ambos casos la
asignacion se hace a la propiedad, usando el método de asignacion, y no directamente a las variables PUsarDispositivo,
PGenerarEventos y PMovimientoMinimo. Esto es asi porque la asignacién de esos valores ha de traducirse en las
correspondientes llamadas a las funciones de la API de Windows.

Inicializacidén de propiedades
Private Sub UserControl_InitProperties|()

UsarDispositivo = PAXJoyl ' Primer dispositivo

GenerarEventos = True ' generando eventos

MovimientoMinimo = 128 ' con un movimiento minimo de 128
End Sub

Lectura de las propiedades
Private Sub UserControl_ReadProperties (PropBag As PropertyBag)
With PropBag

UsarDispositivo = .ReadProperty ("UsarDispositivo", PAXJoyl)

GenerarEventos = .ReadProperty ("GenerarEventos", True)

MovimientoMinimo = .ReadProperty ("MovimientoMinimo", 128)
End With

End Sub

Cuando el control va a ser destruido es preciso guardar los valores de las propiedades, respondiendo adecuadamente al
evento WriteProperties, y eliminar la subclasificacién de la ventana, tarea ésta que llevaremos a cabo en el método
asociado al evento Terminate. Usando la misma funcidn SetWindowLong restituiremos la antigua direccion, tras lo cual
asignamos el valor Nothing a la variable del mddulo que mantenia la referencia al control, asegurdndonos asi su correcta
destruccion. De no efectuar esta asignacién el control no seria destruido, por lo que a pesar de que el programa terminase
el proceso de liberacién no se podria completar satisfactoriamente.

Al destruir el control
Private Sub UserControl_Terminate ()
' restablecemos su procedimiento original
SetWindowLong hWnd, GWL_WNDPROC, Modulo.ProcedimientoAnterior
Set Modulo.ObjetoAsociado = Nothing ' y destruimos la referencia
End Sub

Seleccion del dispositivo a utilizar

Creado el control lo normal es que el siguiente paso sea la seleccién del dispositivo que se va a usar, para lo cual se
asignard el valor adecuado a la propiedad UsarDispositivo. En el método de asignacién de esta propiedad lo primero que
tenemos que hacer es comprobar si el dispositivo que se estaba utilizando hasta el momento estaba "capturado”, caso en
el cual hay que "liberarlo”. Tras esto ya se puede fijar el nuevo dispositivo, que dejaremos en el mismo estado en que se
encontraba el anterior.

Para que una palanca de juegos nos envie mensajes cada vez que se produzca una modificacion en el estado de los
botones o la posicion, es necesario llamar previamente a la funcion joySetCapture. Esta funcidn necesita cuatro
parametros: el identificador de la ventana a la que se han de enviar los mensajes, el identificador del dispositivo a
capturar, un intervalo de tiempo que se usara como periodo de envio de los mensajes y un indicador. Tras llamar a esta
funcién se dice que hemos "capturado” el dispositivo, ya que en cierta forma tomamos en exclusiva su funcionamiento.
De hecho, si un programa hace una segunda llamada a joySetCapture obtendra un error como resultado. Por eso es
necesario liberar el dispositivo en cuanto no lo estemos usando, simplemente llamando a la funcién joyReleaseCapture
pasando como parametro el identificador de la palanca de juegos.

El tercer pardmetro de joySetCapture especifica un tiempo que ha de transcurrir entre el envio de dos mensajes
consecutivos a la ventana. Si pasamos el valor O y como cuarto parametro el valor True, estaremos comunicando a
Windows que no genere un mensaje periddico, sino sélo cuando se haya pulsado o liberado un botén o bien cuando se
registre un movimiento superior a una cierta magnitud. Esta magnitud se establece mediante la funcién joySetThreshold,
que necesita como primer parametro el identificador del dispositivo y como segundo el factor citado.

El codigo correspondiente al método de asignacion a la propiedad UsarDispositivo, que puede ver a continuacion,
efectua todo este proceso sélo si el valor que se va a asignar es diferente al que existe en ese momento, es decir, si
efectivamente se va a cambiar el dispositivo, ya que de lo contrario no tiene sentido realizar el trabajo.

Public Property Let UsarDispositivo (NuevoValor As PAXJoyDispositivo)

If PUsarDispositivo <> NuevoValor Then ' si se modifica realmente
If PGenerarEventos And Ambient.UserMode Then ' si se tiene el dispositivo
joyReleaseCapture PUsarDispositivo
End If
PUsarDispositivo = NuevoValor ' asignamos el nuevo dispositivo
If PGenerarEventos And Ambient.UserMode Then ' si se tenia el dispositivo
joySetCapture hWnd, PUsarDispositivo, 0, True ' dejarlo como estaba
joySetThreshold PUsarDispositivo, PMovimientoMinimo
End If
End If

End Property

Observe que las llamadas a joySetCapture, joySetThresholdy joyReleaseCapture se realizan solo en caso de que el modo
actual sea de ejecucion, ya que no tiene sentido llamarlas cuando el control se encuentra en modo de disefio puesto que
no es posible recuperar informacién alguna.

Las propiedades GenerarEventos y MovimientoMinimo

Los métodos de lectura de estas propiedades, al igual que el de la propiedad UsarDispositive, se limitan a devolver el
valor almacenado en la variable privada correspondiente, sin efectuar ninguna operacion adicional.

Al modificarse la propiedad GenerarEventos habra que capturar el dispositivo, si el nuevo valor es True, o liberarlo, en
caso de que sea False. Ambas operaciones se efectiian solo si estamos en modo de ejecucion.

Public Property Let GenerarEventos (NuevoValor As Boolean)

PGenerarEventos = NuevoValor ' guardamos el valor
If Ambient.UserMode Then ' si estamos en ejecucidn
If PGenerarEventos Then ' si hay que generar eventos
joySetCapture hWnd, PUsarDispositivo, 0, True ' capturar el dispositi
Else ' en caso contrario
joyReleaseCapture PUsarDispositivo ' liberarlo
End If
End If

End Property

De forma anéloga, la asignacién de un cierto valor a la propiedad MovimientoMinimo ha de traducirse en una llamada a
la funcion joySetThreshold, estableciéndose el nuevo movimiento minimo que ha de registrarse antes de que se envie un
mensaje de notificacién.

Public Property Let MovimientoMinimo (NuevoValor As Long)
PMovimientoMinimo = NuevoValor ' guardamos el wvalor
' si estamos ejecutando
If Ambient.UserMode Then _
joySetThreshold PUsarDispositivo, NuevoValor ' lo pasamos al dispositivo
End Property

Propiedades informativas

Seleccionado el dispositivo que se va a utilizar, lo normal sera comprobar si en realidad hay o no conectada una palanca
de juegos. Con este fin se consultard el valor de la propiedad s6lo de lectura EstaConectado. Dicha propiedad ha de
devolver True o False, segiin haya o no un joystick conectado en ese momento. Tendremos que usar dos nuevas
funciones para determinar si un dispositivo estd o no disponible y conectado: joyGetNumDevs y joyGetPos. La primera
de ellas devuelve el nimero de palancas de juegos que es posible conectar de forma simultdnea, por regla general dos. La
segunda recupera el estado actual de los botones y posicién de la palanca, devolviendo el valor JOYERR NOERROR si
no hay ningtin problema.

El ntimero de dispositivo seleccionado, que se almacena en la variable PUsarDispositivo, siempre habré de estar
comprendido entre 0 y el numero devuelto por joyGetNumDevs menos 1. Es decir, el valor que retorna joyGetNumDevs
debe ser mayor que PUsarDispositivo, de lo contrario el dispositivo elegido no serd vélido.

joyGetNumDevs devuelve el nimero de dispositivos que es posible conectar, no el que hay conectado en este momento.
Si tenemos instalado el controlador estandar de palancas de juego de Windows, esta funcidén devolvera el valor 2 a pesar
de que no tengamos ningtn dispositivo conectado. Para comprobar si hay o no conectada una palanca de juegos es
preciso consultarla, preguntandole el estado de los botones y la posicion. Para ello se usa la funcién joyGetPos, pasando
como primer parametro el identificador del dispositivo a consultar y como segundo una variable del tipo JOYINFO. Si la
llamada a joyGetPos devuelve el valor JOYERR NOERROR es que la palanca estd conectada.

' Propiedad sélo de lectura que devuelve True si hay un joystick conectado
Public Property Get EstaConectado() As Boolean
Dim Inf As JOYINFO

' Devolvemos True si joyGetNumDevs devuelve un numero mayor
' que el del dispositivo que queremos usar y, ademds, la
' llamada a joyGetPos no devuelve error alguno
EstaConectado = joyGetNumDevs > PUsarDispositivo And _
joyGetPos (PUsarDispositivo, Inf) = JOYERR_NOERROR
End Property

El resto de las propiedades informativas, todas de sélo lectura, se limitan a recuperar y devolver uno de los datos de la
estructura JOYINFO. Este tipo cuenta con los miembros enumerados en la tabla 8.3, todos ellos de tipo Long. En lugar de
llamar a la funcién joyGetPos en el método de lectura de las propiedades PosX, PosY, PosZ v Botones, codificaremos
un método privado que sera el encargado de realizar esa llamada, devolviendo como resultado la estructura de datos
JOYINFO. De esta forma los métodos de lectura de las propiedades se pueden limitar a devolver el miembro que
corresponda, sin mas.

Tabla 8.3: Miembros del tipo JOYINFO

Nombre Contenido

wXpos Posicién en el eje X
wYpos Posiciénen el eje Y
wZpos Posicién en el eje Z
wButtons Estado de los botones

' Método privado para consultar el estado del dispositivo
Private Function Estado() As JOYINFO
Dim Inf As JOYINFO

joyGetPos PUsarDispositivo, Inf ' consultamos el dispositivo
Estado = Inf ' y devolvemos la informacién
End Function

' Propiedades sélo de lectura para recuperar las coordenadas
' de la palanca de juegos

Public Property Get PosX () As Long
PosX = Estado.wXpos

End Property

Public Property Get PosY () As Long
PosY = Estado.wYpos
End Property

Public Property Get PosZ () As Long
PosZ = Estado.wZpos
End Property

' Propiedad sélo de lectura para recuperar el estado de los botones
Public Property Get Botones () As Long

Botones = Estado.wButtons
End Property

Proceso de los mensajes y generacion de eventos

Suponiendo que la propiedad GenerarEventos del control esté activada, todos los mensajes que Windows envie a
nuestro control pasaran previamente por la funcién ProcesaMensaje, que hemos alojado en un médulo de codigo
estandar. Esta funcién comprobaré si el mensaje que se recibe, en el segundo parametro, es uno de los relacionados con
las palancas de juegos, en cuyo caso lo pasara al método Evento del control. Indistintamente de si se da este paso o no, el
mensaje siempre serd enviado al anterior procedimiento de ventana mediante la funcién CallWindowProc.

' Funcidén que se encargard de procesar los mensajes
Function ProcesoMensaje (ByVal hWnd As Long, ByVal Mensaje As Long, _
ByVal wParam As Long, ByVal lParam As Long) As Long

' Si es uno de los mensajes gque nos interesa
If (Mensaje >= MM_JOY1BUTTONDOWN And Mensaje <= MM_JOY2BUTTONUP) Or _
(Mensaje >= MM_JOY1IMOVE And Mensaje <= MM_JOY2ZMOVE) Then
ObjetoAsociado.Evento Mensaje ' lo notificamos a nuestro cliente
End If

' En cualquier caso llamar a CallWindowProc y devolver
' el valor que obtengamos
ProcesoMensaje = CallWindowProc (ProcedimientoAnterior, hWnd, Mensaje, wParam,
End Function

El método Evento del control PAXJoystick decidira qué evento generar segin el mensaje que se haya recibido. A pesar
de ser publico, este método esta pensado para utilizarse solamente como medio de comunicacién entre el modulo
estandar y el control, por lo que usando la ventana de atributos de procedimiento lo haremos oculto.

' Este procedimiento serd llamado desde el mddulo
' de cédigo para notificar un evento
Public Sub Evento (Mensaje As Long)
Select Case Mensaie
Case MM_JOY1BUTTONDOWN, MM_JOY2BUTTONDOWN
RaiseEvent OnButtonDown
Case MM_JOY1BUTTONUP, MM_JOY2BUTTONUP
RaiseEvent OnButtonUp
Case Else
RaiseEvent OnMove
End Select
End Sub

Con esto ya tenemos terminado nuestro control, el primero que hemos subclasificado. Aunque funcionara correctamente,
lo vamos a ver en un momento, el método de comunicacién usado entre el control v el médulo estandar es muy débil.

Uso de un PAXJoystick

Utilizar un control subclasificado en el interior del entorno de desarrollo de Visual Basic es bastante peligroso. Cualquier
error que nos haga entrar en el modo de interrupcién puede provocar el bloqueo completo del entorno, asegtrese de
guardar siempre antes su trabajo. Una alternativa m4s segura consiste en compilar siempre el cédigo y ejecutarlo desde
fuera del entomo de desarrollo.

Para comprobar el funcionamiento de nuestro control vamos a disefiar un formulario como el que se muestra en la figura
8.5. Tenemos cuatro controles TextBox para mostrar el estado de los botones, otros tres para mostrar la posicion en cada
uno de los ejes, un botdn para activar/desactivar la generaciéon de eventos, otro para actualizar la informacién que hay en
la ventana y, por tltimo, un control PAXJoystick.

im. PPAXJoystick - Formulario [Form]
m] [m]

. Uso del componente PAJoystick lD

. Botén1: .. BotdnZ2. .. Botdn3: .. Botdnd. . .
| | | |
S,
| | | i
Desactivar eventu:usl Refrezcar |
B = ................ L

Figura 8.5: Formulario para mostrar el estado de la palanca de juegos

Al cargarse el formulario se consultara el valor de la propiedad EstaConectado, notificando si hay o no conectada una
palanca de juegos. En caso negativo el programa termina directamente. Si tenemos un joystick conectado a la salida
elegida en la propiedad UsarDispositivo, cada vez que se pulse el botéon Refrescar recuperaremos los datos de las
propiedades informativas. El estado de los botones lo indicaremos cambiando el color de fondo de los TextBox, que sera
rojo si el botédn correspondiente estd pulsado o blanco en caso contrario. Las posiciones en los distintos ejes se mostraran
directamente, asignando el valor de las propiedades PosX, PosY y PosZ a la propiedad 7ext de las cajas de texto.

La pulsacién del botén que hay a la izquierda activard/desactivard la generacién de eventos por parte del control, para lo
cual invertird el estado de la propiedad GenerarEventos. Segun el estado actual, el botén mostrara un titulo u otro. En
los métodos asociados a los eventos OnButtonDown, OnButtonUp y OnMove simplemente llamaremos al método
correspondiente al evento Click del boton Refrescar, actualizando la informacion mostrada.

Como puede ver en el siguiente fragmento, para comprobar el estado de los botones no se han usado las constantes que
los representan. En su lugar se ha utilizado un bucle cuyo contador sirve como indice de una operacién de potenciacién
cuya base es 2. El resultado es exactamente el mismo, puesto que los valores obtenidos (1, 2, 4 y 8) son los que
representan esas constantes, sin embargo el cédigo es més corto.

Cada vez que se pulse el botdén Refrescar
Private Sub Commandl_Click ()
Dim Contador As Integer

With PAXJoystickl ' usamos el control PAXJoystick
For Contador = 0 To 3 ' para recuperar el estado de los cuatro botones

tbBoton (Contador) .BackColor = _
ITf((.Botones And 2 ~ Contador) <> 0, vbRed, vbWhite)

Next
tbX = .PosX ' y las posiciones
tbY = .PosY
tbZ = .PosZ
End With

End Sub

' La pulsacidén del botdn Activar/Desactivar eventos
Private Sub Command2_Click ()
With PAXJoystickl
' invertimos el estado de la propiedad GenerarEventos

.GenerarEventos = Not .GenerarEventos
' seglin el estado actual fijamos el titulo del botdn
Command?2.Caption = IIf(.GenerarEventos, "Desactivar eventos", "Activar ev
End With

End Sub

' Al cargar el formulario
Private Sub Form_Load()
' Si hay una palanca de juegos
If PAXJoystickl.EstaConectado Then
' lo notificamos
MsgBox "Hay una palanca de juegos conectada"
Else ' en caso contrario avisamos
MsgBox "No hay una palanca de juegos conectada"
Unload Me ' y descargamos
End If
End Sub

' Cada vez que se produzca un evento
Private Sub PAXJoystickl_OnButtonDown ()
Commandl_Click
End Sub

Si ejecuta el programa y no dispone de una palanca de juegos conectada a su sistema no podra ver resultado alguno. En
la figura 8.6 puede ver el formulario mostrando el estado actual de los botones y la posicién en los ejes X e Y.

im. Uso del componente PAS) oystick I
Botdn 1 Botan 2 Botdn 3 Botdn 4
s Y
|32844 |78 i
Desactivar eventos | Refrezcar

Figura 8.6: El programa PPAXJoystick en funcionamiento
La importancia de la encapsulacion

El disefio del control que acabamos de crear cuenta con el mismo problema que teniamos inicialmente con el control
PAXEnumeraVentanas, la falta de encapsulacion del cédigo alojado en el médulo estandar. Este problema puede ser el
origen de diversos problemas que nos puede llevar mucho tiempo descubrir, sobre todo si no somos conscientes de que
las variables globales de un componente ActiveX son compartidas por las diferentes copias del objeto.

Supongamos que inserta en un formulario un control PAXJoystick. Justo en ese momento se produce la inicializacion,
en la cual se asigna la direccion del anterior procedimiento de ventana a la variable ProcedimientoAnterior, ademas de
asignarse a ObjetoAsociado una referencia al propio control. Estas dos variables son globales, lo cual quiere decir que

son compartidas. Si ahora inserta en el formulario un segundo control PAXJoystick, de nuevo se asigna a
ProcedimientoAnterior la antigua direccion del procedimiento de ventana, perdiéndose el anterior contenido de esta
variable. La variable ObjetoAsociado, por su parte, ahora apuntara al segundo control.

En este punto nos encontramos con que la funcién que hay en el médulo estandar esta recibiendo mensajes de los dos
controles insertados, sin embargo en la llamada a CalllWindowProc todos los mensajes se pasan al procedimiento cuya
direccién se almacena en ProcedimientoAnterior, es decir, al procedimiento de ventana del segundo control insertado.
De igual forma, cuando se encuentra un mensaje relativo a la palanca de juegos se notifica al control cuya referencia se
almacena en ObjetoAsociado, que también es el segundo. El primer control deja de funcionar por completo.

La situacién descrita, sin embargo, puede no ser la peor. Partiendo del escenario anterior imagine que eliminamos el
segundo de los controles. En ese momento se restituye la antigua direccién del procedimiento de ventana del control, tras
lo cual se asigna el valor Nothing a la variable ObjetoAsociado. En este momento la funcién ProcesaMensaje atin
puede recibir mensajes del primer control, puesto que éste sigue subclasificado. Dichos mensajes serdn pasados mediante
CallWindowProc a la funcién cuya direccion se encuentra en ProcedimientoAnterior, es decir, a la direccidn del
procedimiento de ventana original del segundo control, que en ese momento ya no existe. Si el mensaje recibido notifica
un evento de la palanca de juegos, se intentard pasar al control asociado usando la referencia almacenada en
ObjetoAsociado, pero en este momento dicha variable contiene el valor Nothing. Como puede imaginarse, el resultado
es desastroso.

La solucidn a estos problemas es clara: necesitamos encapsular las variables publicas del modulo, de tal forma que cada
control cuente con las suyas privadas. La forma de conseguir esto, sin embargo, no est4 tan clara. Al disefiar el control
PAXEnumeraVentanas encontramos una manera de hacerlo bastante eficiente, pero que no es vélida en este caso. La
funcion ProcesaMensaje, que es a la que Windows llama directamente, no cuenta con un parametro adicional facilitado
por nosotros, como sf ocurria con la funcién a la que llamaba EnumWindows.

Datos de usuario en una ventana

Anteriormente describiamos una ventana como una estructura de datos, en la cual existen miembros que establecen el
estilo de la ventana, el procedimiento de gestién de mensajes y otros elementos. En dicha estructura existe un espacio
reservado para datos de usuario, espacio que esta totalmente a nuestra disposicion.

Para guardar un valor en ese espacio usaremos la funcion SetWindowLong que ya conocemos, facilitando como segundo
pardmetro la constante GWIL_USERDATA. Con el fin de recuperar el dato, cuando nos interese, usaremos la funcién
GetWindowLong, que tomando como parametros el identificador de ventana y la citada constante devuelve el valor
almacenado.

El espacio reservado para datos de usuario en la estructura de una ventana es de sélo cuatro bytes, es decir, un valor de
tipo Long. Podriamos usar este espacio para almacenar, por ejemplo, la direcciéon del antiguo procedimiento de ventana
que hasta ahora guarddbamos en la variable ProcedimientoAnterior. Sin embargo seguiriamos teniendo el problema de
la referencia al control, conservada en la variable ObjetoAsociado. En realidad tenemos dos problemas: necesitamos
almacenar mas datos de los que podemos, por un parte, y queremos guardar una referencia a un objeto en un espacio
pensado para un Long.

La solucién a estos problemas la encontraremos en el uso de punteros, tal y como vamos a ver en el punto siguiente. No
se trata de la técnica mas facil pero seguramente si de la mas efectiva.

Punteros a variables, cadenas y objetos

(Es necesario el uso de punteros para programar? Si hace esta pregunta a un grupo de programadores seguramente
obtendra multitud de respuestas, pero la mayoria de ellas se podrdn agrupar en dos categorias: "El uso de punteros es
fundamental para la programacién” y "Hay que evitar por todos los medios el uso de punteros, es una aberraciéon” La
primera respuesta la obtendra de aquellos programadores que han utilizado punteros siempre, 1€éase aquellos que desde
sus inicios han usado lenguajes como C/C++. La segunda es propia de "puristas" que consideran que trabajar con
direcciones de memoria en un programa es algo que deberia estar prohibido, por los riesgos que conlleva, los problemas
de portabilidad, etc.

;Qué respuesta darian los programadores Visual Basic a esa pregunta? Seguramente una contestacion resignada: "En
Visual Basic no es posible trabajar con punteros, una lastima"; o bien una respuesta de satisfaccion: "Por suerte en Visual
Basic no existen los punteros" Desde luego Visual Basic no es un lenguaje disefiado para trabajar de forma cémoda con
punteros pero, por suerte, no es algo imposible. Si piensa que, en ocasiones, el uso de los punteros puede estar justificado
para conseguir una cierto fin esta de suerte, porque en Visual Basic 5 es posible trabajar con punteros.

Existen tres funciones de las cuales no encontrara documentacién en los manuales de Visual Basic, ni en la ayuda en
linea. Ni siquiera podra encontrarlas usando el Examinador de objetos, puesto que estan contenidas en un moédulo oculto.
Si puede encontrarlas, sin embargo, usando la herramienta OLEVIEW que utilizamos en un capitulo anterior para
1nspeccwnar el contenido de las librerias de tipos. Cargando la libreria de tipos del archivo VBAS.DLL, que encontrara
en el propio directorio de Visual Basic, podra ver un moédulo llamado HiddenModule. Las funciones que nos interesan
son las tltimas tres de este médulo, destacadas en la imagen de la figura 8.7. Sus nombres son VarPtr, StrPtry ObjPtr.

i |TypeLib Yiewer _ |&]

File  “Wiew

B- ﬂ YBA [Visual B asic For Applicatior = | BSTRE _=stdcall _B =str InputB( :_j
-2 Enums [in] long Humber.
}' el [in] short FileHumber):
: IS 5 T [entry(0=60000002), helpcontext(0=000f735f)]
=L Modules VARIANT _stdcall _B_war_ InputB(
- Bl module Constants [in] long Humber,
E e SOk e [ (0 snnnnnngin]hs?nrt Flle}glnlmgﬁﬂ 735817
R . entry(l= . helpcontextilx
Eel-gifl, module Convarsion PETR. mpdeal) B ate Tnpit
I'_:'J"'HL?E. rmodule FlleSys.tem Fin] ong Nunbers
EJ"'HL?E- module D ateTime [in] short FilsHumber):
T module Infarmation [entry (0=60000004), helpcontext{0=z000£f735f)]
'L’:;J""HL?@ il nter actics VARIANT _=tdcall _.E_var_Input(
dule Math [in] long Humnber,
EJ"'HL?E' e .at . [in] short FileNumber):
e module Financial [entry(0=60000005), helpcontext(0=000f65a4)]
=Tl module _Hiddentadule void _stdeall Width(
(@8 Funct [1n] zhort FileHumber.
E: “_j-_!__":mi"m short Width):
fray! ntry{0=x6 000000 ). hidden]
""" m _EB_str_lnputb g ([in] woid* Ptr):
------ m B war |nputB hidde:
------ m B str_Input FPtri{[in] BSTRE Ptr):
...... B 5 hlljljn—'n]
P m ;\?ia:fhar_lnput . LT ITUnknown* Ptr):
S m arPh
M SuPly s
- OhiPtr
|-;-_ A Intarfares = pa—
« | LI_I =

Figura 8.7: Contenido del médulo oculto existente en la libreria VBAS.DLL

Si escribe en el Editor de codigo el nombre de cualquiera de estas tres funciones y pulsa un espacio, vera que aparece una
indicacién de los parametros que es necesario pasar. Esto nos demuestra que las funciones son reconocidas por Visual
Basic, de lo contrario se las trataria como a cualquier otro identificador desconocido.

La funcién VarPtr

Esta funcion no le resultard nueva si lleva muchos afios programando con BASIC. VarPtr, usada conjuntamente con
Peek 'y Poke, permitia a los programadores de entonces efectuar operaciones de bajo nivel, sin las cuales no era posible
conseguir muchos resultados. Se podia, por ¢ emplo declarar una matriz de bytes, escribir en ella con Poke y después
ejecutarla como un programa en cédigo méquina obteniendo la direccién de la matriz con VarPtr.

La funcion VarPtr acepta como parametro una variable de cualquier tipo, devolviendo un entero Long con la direccién
de memoria en la que se almacena el contenido de dicha variable. Esta direccién puede ser almacenada en otra variable,
pasada como pardmetro a una funcion o usada en cualquier otro contexto en la que se puede utilizar un Long.

(Cuéando necesitara usar la funcion VarPtr? En caso de que haya que pasar a una funcién de la API de Windows la
direccién de una variable, siempre puede declarar el parametro correspondiente como ByRef. Algunas funciones, sin
embargo, necesitan como parametro una estructura de datos en cuyo interior existe punteros a variables u otras
estructuras. En casos asi VarPtr facilitara el trabajo, ya que bastard con realizar una asignacion como la siguiente:

Estructura.Puntero = VarPtr (Variable)

Este es uno de los pocos casos en que encontraremos utilidad a VarPtr. La manipulacion del contenido de las variables, a
partir de la direccién facilitada, siempre quedard en manos de la funcién de la API a la que llamemos. Aunque hay pocos

casos que lo justifiquen, también desde Visual Basic es posible alterar el contenido de esa zona de memoria, como
veremos en un punto posterior.

La funcién StrPtr

El almacenamiento de las cadenas en Visual Basic no es tan sencillo como cabria esperar, a diferencia de lo que ocurre
en otros lenguajes en los que una cadena es, simplemente, una secuencia de bytes consecutivos. Cuando declaramos una
variable de tipo Strmg lo que se aloja en ella es un puntero, inicialmente nulo. Al asignarle un valor, Visual Basic asigna
un bloque de memoria y copia en €l la cadena, guardando la direccién de dicho bloque en la variable String. El bloque
que contiene la cadena también almacena, al inicio, un entero Long que indica la longitud, asi como un caracter nulo al
final.

La funcion VarPtr devuelve la direccion de una variable, lo que nos permite saber la posicién de memoria en que esta
almacenado el valor que contiene. El contenido de una variable String es a su vez un puntero, por lo que para obtener la
direccién real de la secuencia de caracteres que forman la cadena deberemos usar la funcion St»Ptr. Puede comprobar la
diferencia existente entre los valores devueltos por VarPtr y StrPtr escribiendo el codigo siguiente:

Dim Cadenal As String

Dim Cadena2 As String

Cadenal = "Hola"

Print "Cadenal -> ", VarPtr (Cadenal), StrPtr (Cadenal)
Print "Cadena2 -> ", VarPtr (Cadena2), StrPtr (Cadena?2)

En la figura 8.8 puede ver que los valores devueltos por VarPtr son dos direcciones consecutivas, que en su sistema
pueden ser, de hecho es lo normal, otras totalmente diferentes. La primera cadena contiene actualmente un valor, por lo
que StrPtr devuelve la direccién en que se encuentra la secuencia de caracteres. La segunda, por el contrario, estd vacia,
de ahi que la direccidn devuelta por StrPtr sea nula.

. Punteroz a cadenas | _ (O] x|

Cadenal -» 1242584 1622732
Cadenad -» 1242580 0

Figura 8.8: Direcciones de cadenas obtenidas con VarPtry StrPtr

Internamente Visual Basic almacena las cadenas en formato Unicode, esto es, usa 16 bits para cada caracter, no 8. Desde
la versién 4.0 Visual Basic estd basado principalmente en COM, todos sus objetos y controles son objetos COM. La
especificacién COM establece que todas las cadenas habran de ser Unicode, no existiendo la posibilidad de trabajar con
cadenas ANSI. Visual Basic se encarga de realizar la conversién siempre que es necesario, pero al acceder directamente
a la cadena esta conversion no se efectia, hemos de tenerlo en cuenta.

La funcién ObjPtr

Desde el mismo momento en que comenzamos a trabajar con Visual Basic estamos usando objetos, como el formulario,
los controles, las colecciones y nuestros propios componentes. Sabemos que estos elementos son objetos COM, de los
cuales podemos tener referencias en variables del tipo correspondlente Lo que se almacena en esas variables, la propia
referencia, no es mas que una direccién, un apuntador a la posiciéon de memoria en la que se encuentra el objeto
propiamente dicho, sus interfaces y su codigo.

Al igual que ocurre con las cadenas, la funcién VarPtr devuelve la direccion de la variable que contiene la referencia,
pero no la direccion del objeto en si. Para conseguir este dato deberemos usar la funcién ObjPtr. Puede hacer una prueba
usando el siguiente co6digo, en el que se asignan diferentes referencias de un mismo objeto a varias variables.

Dim ObjetoOriginal As New Collection
Dim ObjetoCopia As Collection
Dim ObjetoGenerico As Object

Print "ObjetoOriginal -> ", VarPtr (ObjetoOriginal), ObjPtr (ObjetoOriginal)
Print "ObjetoCopia -> ", VarPtr (ObjetoCopia), ObjPtr (ObjetoCopia)

Print "ObjetoGenerico -> ", VarPtr (ObjetoGenerico), ObjPtr (ObjetoGenerico)
Set ObjetoCopia = ObjetoOriginal

Set ObjetoGenerico = ObjetoOriginal

Print "ObjetoCopia/Set-> ", VarPtr (ObjetoCopia), ObjPtr (ObjetoCopia)

Print "ObjetoGenerico/Set -> ", VarPtr (ObjetoGenerico), ObjPtr (ObjetoGenerico

Légicamente cada variable es independiente de las demas, por lo cual tiene su propia direccion. Esta la obtenemos con la
funcién VarPtr. Inicialmente ObjetoCopia y ObjetoGenerico no contienen referencia alguna, por ello ObjPtr devuelve
el valor 0. Tras una asignacién podemos comprobar que las tres referencias contienen la direccion de un mismo objeto.

im. Punteros a objetos |
O bjetalriginal - 1242584 1713616
ObjetoCopia  -» 1242580 ]
ObjetoGenerico -» 1242576 ]
ObjetaCopias/Set-» 1242580 1713616
Objetol enenicodSer -» 1242576 1713616

Figura 8.9: Direcciones de objetos devueltas por VarPtry ObjPtr

Usando la funcién ObjPtr podemos guardar la direccion de un objeto en la estructura de datos de una ventana, usando la
ya conocida SetWindowLong. Con esto solucionamos uno de los problemas indicados anteriormente, la imposibilidad de

guardar una referencia a un objeto. Seguimos teniendo, no obstante, un problema adicional: s6élo podemos guardar un
dato de 4 bytes asociado a la ventana.

Manipulacion directa de la memoria

Las tres funciones que hemos conocido en los puntos anteriores nos permiten obtener la direccidén de memoria en la que
se encuentra el contenido de cualquier variable, indistintamente de que sea de un tipo numérico, una cadena o un objeto.
Con esto s6lo podemos recuperar un puntero y usarlo como parametro, bien sea para pasarlo a una funcién de la API de
Windows o para guardarlo en una variable. No podemos, sin embargo, manipular directamente el contenido de ese
espacio de memoria. No, en Visual Basic no existen las anteriormente mencionadas Peek y Poke, con las cuales era
posible leer o escribir un byte en memoria.

Buscando en las diferentes librerias de Windows, sin embargo, podemos encontrar algunas funciones utiles, entre ellas
una llamada RtiMoveMemory. Este procedimiento facilita la copia de una zona de memoria a otra, para lo cual es
necesario facilitar tres pardmetros: direccién de destino, direccién de origen y numero de bytes a copiar. Podemos usar
esta funcion en Visual Basic declarandola de 1a siguiente forma:

Private Declare Sub RtlMoveMemory Lib "kernel32" _
(ByVal DestAddr As Long, ByVal SrcAddr As Long, ByVal Bytes As Long)

Los valores devueltos por VarPtr, StrPtr'y ObjPtr, punteros en los tres casos, no son peligrosos por si solos, pero usados
conjuntamente con Rt/MoveMemory pueden ser muy utiles al tiempo que desastrosos. Un fallo insignificante puede
causar el bloqueo del programa o el sistema, asi de sencillo. En el siguiente codigo, por ejemplo, se copia el contenido de
una variable de tipo entero a otra del mismo tipo. ;Qué ocurriria si la longitud, el niimero de bytes a copiar, fuese
erroneo? Se escribiria en zonas de memoria que nada tienen que ver con la variable, algo realmente peligroso. En este

caso podré ver que tras usar Rt/MoveMemory ambas variables contienen el mismo valor, es como si se hubiese efectuado
una asignacion.

Dim Numerol As Integer, Numero2 As Integer
Numerol = 1234
Print "Antes - —————————————— "
Print "Numerol -> " & Numerol
Print "Numero2 -> " & Numero2
Rt1MoveMemory VarPtr (Numero2), VarPtr (Numerol), Len (Numerol)
Print "Después ——————————————— "
Print "Numerol -> " & Numerol
Print "Numero2 -> " & Numero2

Esté4 claro que para asignar una variable de tipo entero a otra no es necesario usar RtiMoveMemory, esto es sélo un
ejemplo. En la practica, no obstante, nos podemos encontrar con casos en los que no existe otra alternativa. Suponga que
al subclasificar un control quiere almacenar en la estructura de datos de ventana un puntero al propio componente. No
tiene mas que recuperar dicho puntero, con ObjPtr, y después copiarlo con SetWindowLong. En el procedimiento de
ventana, que gestionard los mensajes, podria recuperar la direccién del control con GetWindowLong pero, jcémo lo
puede usar? Lo que obtiene es un puntero, un valor de tipo Long, no una referencia a un objeto. Una asignacion directa
del puntero a una variable que espera una referencia no es posible, pero si se puede conseguir usando RtlAMoveMemory.

Un mejor control PAXJoystick

Veamos cémo podemos aprovechar lo que hemos aprendido para mejorar el disefio de nuestro control PAXJoystick,
encapsulando el funcionamiento del médulo estandar para evitar problemas cuando se usan varios controles de forma
simultanea.

Puesto que necesitamos almacenar dos datos: la referencia al objeto que es el control y la anterior direccién del
procedimiento de ventana, no podremos alojarlos directamente en la estructura de datos de la ventana, ya que en ella sélo
hay espacio para 32 bits. Si podemos, sin embargo, almacenar los dos datos en una estructura y guardar la direccion de
ésta con SetlWindowLong. Esta es la idea que vamos a desarrollar.

Almacenamiento de la informacion de enlace

Comenzaremos por definir la estructura de datos que vamos a utilizar, un nuevo tipo al que llamaremos PAXEnlace.
Puesto que va a ser necesario usarlo tanto en el control como en el médulo estandar, incluiremos la definicién en este
ultimo. Este nuevo tipo tendra dos miembros de tipo Long: ControlAsociado y ProcedimientoAnterior. El primero
tendr4 la direccion del control PAXJoystick2 asociado, mientras que el segundo almacenard la direccién del anterior
procedimiento de ventana.

En el médulo de cédigo del control incluiremos una nueva variable privada, llamada Enlace, de tipo PAXEnlace. Esta
variable no va a ser accesible de ninguna forma para el usuario final del control, nos servira internamente para guardar la
informacién de enlace entre el médulo estandar y el control mismo. Al producirse el evento de inicializacién, momento
en el que subclasificaremos, guardaremos en la estructura de la ventana la direccién de la variable Enlace, que como
puede ver obtenemos mediante la funcion VarPtr. Tras esto recuperamos la direccion del control, con la funcién ObjPtr,
y la guardamos en el miembro ControlAsociado. Por tltimo modificamos el procedimiento de ventana, guardando la
direccién del anterior en ProcedimientoAnterior.

' Al inicializar el control
Private Sub UserControl_Initialize()
' guardamos como dato de usuario la direccidn
' de la variable Enlace
SetWindowLong hwnd, GWL_USERDATA, VarPtr (Enlace)
Enlace.ControlAsociado = ObjPtr (Me)
Enlace.ProcedimientoAnterior = _

SetWindowLong (hwnd, GWL_WNDPROC, AddressOf Modulo.ProcesoMensaije)
End Sub

Observe que lo que guardamos en la informacion de enlace no es una referencia al objeto, sino la direccion de éste. Dicha
direccién no se puede usar directamente, sera necesario reconstruir el objeto a partir de ella.

Cuando el control se destruya deberemos restituir el anterior procedimiento de ventana, para lo cual bastara con llamar a
SetWindowLong pasando el miembro ProcedimientoAnterior como tltimo parametro.

' Al destruir el control
Private Sub UserControl_Terminate ()
' restablecemos su procedimiento original
SetWindowLong hwnd, GWL_WNDPROC, Enlace.ProcedimientoAnterior
End Sub

Recuperacion de los datos en el procedimiento de ventana

Una vez subclasificado el control, los mensajes comenzaran a llegar a la funcién ProcesoMensaje del moédulo estandar.
Es en este punto donde necesitamos recuperar la informacién de enlace. Para ello primero obtenemos la direccién de la
estructura PAXEnlace usando la funcion GetWindowLong. La direccion obtenida la usamos con el procedimiento
RtiMoveMemory para copiar los datos a una variable privada que hemos declarado, como puede verse en el siguiente
fragmento.

Private Declare Sub RtlMoveMemory Lib "kernel32" _

(DestAddr As Any, SrcAddr As Any, ByVal Bytes As Long)

Dim Direccion As Long ' recuperamos la direccidn

Direccion = GetWindowLong (hwnd, GWL_USERDATA) ' de los datos de enlace

Dim Enlace As PAXEnlace ' variable de enlace vacia
' recuperamos la informacién original
Rt1lMoveMemory Enlace, ByVal Direccion, 8

Observe que se ha modificado la definicion de la funcion RtiMoveMemory, de tal forma que los dos primeros parametros
se reciben por referencia. Esto evita tener que estar usando continuamente la funcién VarPtr para obtener la direccion y
pasarla por valor. Al copiar los datos de enlace, sin embargo, es preciso pasar por valor la direccidn que se ha recuperado
con GetWindowLong, para lo cual simplemente se precede la variable Direccion de la palabra clave ByVal.

En este momento la variable privada Enlace ya tiene la direccion del control y la direccién del anterior procedimiento de
ventana. En realidad lo que hemos hecho es copiar la variable privada Enlace declarada en el control, a nuestra propia
variable privada. En caso de que el mensaje que se recibe esté relacionado con la palanca de juegos es necesario llamar al
método Evento. Antes, sin embargo, deberemos reconstruir el objeto PAXJoystick2 a partir de la direccion que hay en
la variable ControlAsociado. Con este fin declaramos una variable PAXJoystick2 capaz de almacenar una referencia,
tras lo cual usamos Rt/MoveMemory para copiar la direcciéon en su interior. A partir de este momento la variable apunta
al control, por lo que podemos llamar sin ningtin problema a sus métodos.

' Si es uno de los mensajes gue nos interesa
If (Mensaje >= MM_JOY1BUTTONDOWN And Mensaje <= MM_JOY2BUTTONUP) Or _
(Mensaje >= MM_JOY1IMOVE And Mensaje <= MM_JOY2ZMOVE) Then
Dim Objeto As PAXJoystick2 ' Variable sin ninguna referencia
' recuperamos la referencia al objeto original
Rt1MoveMemory Objeto, Enlace.ControlAsociado, 4

Objeto.Evento Mensaje ' lo notificamos a nuestro cliente
Rt1lMoveMemory Objeto, 0&, 4 ' eliminamos el puntero
End If

Objetos sin referencia

La variable Objeto declarada como local en la funcion ProcesoMensaje no tiene inicialmente referencia alguna, por lo
que al llegar al final del c6digo, momento en que se liberan las variables locales, Visual Basic sabe que soélo tiene que
eliminar la variable. Si creamos un objeto y asociamos su referencia a esta variable, al salir de la funcién es necesario
primero destruir el objeto, tras lo cual se elimina la variable. En un capitulo previo analizamos el mecanismo de control
de referencias, consistente en un contador interno que se incrementa con cada asignacién y se decrementa de forma
automatica cuando se eliminan referencias, de tal forma que si el contador llega al valor cero se destruye el objeto.

En el cédigo anterior asignamos a la variable Objeto una referencia obtenida mediante un puntero. No se ha usado la
instruccidn Set, por lo cual el contador de referencias a ese objeto no se ha incrementado. Al llegar al final de la funcion,
sin embargo, puesto que Objeto tiene una referencia, Visual Basic decrementard el contador y destruira el objeto, es
decir, nuestro control, que en este momento se estd ejecutando. Un segundo paso por la funcién de proceso de mensajes
sera an peor, puesto que al salir de ella se decrementara de nuevo el contador y se intentara destruir un objeto que ya no
existe.

Para evitar estos problemas lo que tenemos que hacer es eliminar la referencia de la variable Objeto antes de llegar al
final de la funcién. Por eso hacemos una nueva llamada a Rt/MoveMemory, copiando en esa variable el valor cero.

Terminada la segunda versién del control puede insertarlo en un formulario y comprobar su correcto funcionamiento.
Realmente en este caso sélo notaria la diferencia entre PAXJoystick y PAXJoystick2 insertando dos controles en un
formulario y usando dos palancas de juegos. El control PAXJoystick no seria capaz de funcionar correctamente,
mientras que PAXJoystick?2 si.

Miembros Friend

Otra de las novedades de la version 5.0 de Visual Basic es la posibilidad de definir métodos Friend en el interior de
modulos de clase y controles. La palabra Friend es un calificador como Private y Public, que afecta al &mbito en que es
reconocido el método. Al declarar un método como Friend conseguiremos que sea publico para todo el proyecto, pero
privado para el exterior, es decir, una combinacion entre los dos calificadores existentes hasta ahora.

En ocasiones es necesario usar desde un modulo de un proyecto métodos que se encuentran en otro. Si hacemos esos
métodos publicos los podra usar cualquiera, pero si los hacemos privados no se podran usar nada mas que desde el

propio modulo. El control PAXJoystick cuenta con un método, llamado Evento, que se ha hecho publico por la
necesidad de ser llamado desde el médulo estandar, pero al tiempo se ha modificado uno de sus atributos para que
permanezca oculto, puesto que no es un método candidato a ser usado por el usuario final del control.

Cuando nos encontramos con un caso asi es mucho mejor definir el método como Friend. Cambiando Public por Friend
en la declaracién del método Evento, por ejemplo, conseguiremos que pueda ser usado por el maédulo estandar pero, al
tiempo, evitaremos que sea conocido fuera del proyecto, por lo que el usuario del control no podra utilizarlo.

En otros lenguajes, como C++, también existe el calificador friend, aunque su comportamiento es diferente. Cuando se
declara una clase, por ejemplo, se puede indicar qué otra clase o un cierto método tiene acceso a los miembros de la clase
que se esté declarando. Es decir, no se permite a todo el proyecto tener acceso, lo cual nos permite controlar
perfectamente qué cddigo podra manipular el objeto. Desgraciadamente, en Visual Basic cuando se declara un método
como Friend no es posible establecer limitaciones, el método es visible para todo el proyecto, sin excepcion. Esto, sin
embargo, no serd un problema si todos los médulos del proyecto estan estrechamente relacionados, que es lo habitual.

Resumiendo

Llegados al fin de este capitulo ya estamos preparados para comenzar a desarrollar controles de todo tipo. No sélo
sabemos cémo definir propiedades de diferentes tipos, métodos y eventos, no solamente podemos crear controles que
sepan como comportarse segin el ambiente en el que estdn, ademas hemos aprendido a crear nuestras propias
colecciones, a aprovechar cualquier funcién de la API de Windows que necesite punteros, a subclasificar nuestros
controles, etc. A partir del préoximo capitulo iremos poniendo en practica todo lo aprendido con el fin de crear controles
utiles, lo cual nos servird, ademas, para conocer también nuevas funciones de la APT de Windows y nuevos conceptos.

